
SR50A Series Recommended Measurement  

Interval & Processing 
 
Discussion: 
 
There are scenarios where the SR50A can produce values with higher than expected errors.  For 
example in really low density snow very little echo is returned back to the sensor.  The increase in 
Echo quality numbers is an indication of the weak signals.  Under these circumstances an SR50A  can 
under estimate or over estimate the snow depth.  If the signal is too weak the sensor will also output 
a value of 0 for the Distance to target.  When the echoes are weak the sensors also automatically 
increases its sensitivity.  This makes the sensors more prone to the occasional erroneous reading 
from flying debris, drifting snow or mounting hardware just outside the beam angle. 
 
The reason not to average values is that occasionally a number with a very high error value is 
produced.  This number should be ignored and not averaged.  Based on experience the best 
technique to eliminate errors and filter out high error readings is to take the median value.    This 
technique also helps to automatically filter out the zero readings that can occasionally be produced.  

 
Example 1: 
 
For a given station a reading is taken every 5 seconds for 1 minute and the median value is taken 
from the readings.   
 
If 11 consecutive readings are as such for 
snow depth: 
 
0.33 
0.34 
0.35 
-1.1 (erroneous reading) 
2.0 (erroneous reading) 
0.37 
0.28 
0.36 
0.10 (High error value) 
0.33 
0.32 
 

After being sorted from low to high: 
 
 
-1.1 
0.10 
0.28 
0.32 
0.33 
0.33 
0.34 
0.35 
0.36 
0.37 
2.0 
 

 
The best course of action would be to ignore the 5 lowest values and take the 6th value (0.33). 

Datalogger Program Example 1:  

 
The CR1000 can be programmed with the SortSpa Instruction which will sort your array of 11 (in this 
example) from low to high and then the middle (median) value is selected.   

 
'CR1000 Series Datalogger 
'SR50AT 
 
'Wiring SR50AT 
'Red - 12V 
'Black, White and Clear G - Ground 
'Green SDI-12 - C3 
 
'Declare Public Variables 
Public PTemp, batt_volt 
 
'Once the SR50AT is installed the Distance from the SR50AT to the 
'Ground must be placed into this parameter.  Once successfully transferred 
'to the SR50AT the SR50AT will store the parameter in EE memory and retain 
'the value even when power is removed 
Public NewDistanceToGround As Float 
Public SR50ADistanceToGround As Float 
 
Public SR50AReturnValues(11,3) As Float 
 
Public SR50A_SnowDepth_Meters As Float 
Public SR50A_QualityVal As Float 
Public SR50A_AirTempC As Float 
 
Public XtendedDistValStr As String * 16 
Public ExtendedCMDResult As Float 
Public SR50ADistSort(11) As Float 
Public SR50ASort(11) As Float 
 
Public SDI12commandstring As String * 16 
Public SR50ATMeasureFlag As Boolean 
 
'Declare Other Variables 
'Example: 
'Dim Counter 
Dim SR50Acount As Float 
 
'Define Data Tables 
DataTable (Diagnostics,1,-1) 
  DataInterval (0,1440,Min,10) 
  Minimum (1,batt_volt,FP2,0,False) 
  Maximum (1,batt_volt,FP2,0,False) 
  Minimum (1,PTemp,FP2,0,False) 
  Maximum (1,PTemp,FP2,0,False) 

EndTable 
'Define Data Tables 
DataTable (SR50AT,1,-1) 
  DataInterval (0,60,Min,10) 
  Minimum (1,batt_volt,FP2,0,False) 
  Sample (1,SR50A_SnowDepth_Meters,IEEE4) 
  Sample (1,SR50A_QualityVal,FP2) 
  Sample (1,SR50A_AirTempC,IEEE4) 
EndTable 
 
'Main Program 
BeginProg 
 
  Scan (1,Min,7,0) 
    PanelTemp (PTemp,250) 
    Battery (batt_volt) 
 
    'Once installed enter the actual distance from the SR50AT to the Ground with 
    'no snow present.  If in doubt it is better to use a slightly larger value 
    'than a smaller value.  Errors in the value will show up as an offset error 
    'in the snow depth values 
    'If a new value for the variable NewDistanceToGround is entered (non zero) 
    'The following code will send that value to the SR50AT. 
    If NewDistanceToGround > 0.0 Then 
 
      'Convert the floating point value to a text string for the SDI-12 command 
      XtendedDistValStr = FormatFloat (NewDistanceToGround,"%4.3f") 
 
      'To send the Distance to ground (in meters) value to the SR50A the 
      'SDI-12 extended command is as follows: 
      'aXDM.MMM!- where M.MMM is the value such as 2.345 Meters 
      SDI12commandstring = "XM;" + XtendedDistValStr + "!" 
 
      'Send out the SDI-12 command to the sensor 
      SDI12Recorder (ExtendedCMDResult,3,0,SDI12commandstring,1.0,0) 
 
      'If the Sensor is present and a valid value was received the 
      'SR50A will return a value of 1 
      If ExtendedCMDResult = 1.0 Then 
        'Reset the value so that no more attempts will be made to update the 
        'Distance to Ground value. 
        NewDistanceToGround = 0.0 
 
        'Read back the Distance to Ground value in the SR50A to confirm 
        'This is not necessary but recommended 
        SDI12Recorder (SR50ADistanceToGround,3,0,"R0!",1.0,0) 
        'If the Sensor is present and a valid value was received the 
        'SR50A will return a value of 1 
 
      EndIf 
    EndIf 

  'After a reset the SR50ADistanceToGround value will be zero.   
    If SR50ADistanceToGround = 0.0 Then 
      'Read back the Distance to Ground value in the SR50A to confirm 
      'This is not necessary but recommended 
      SDI12Recorder (SR50ADistanceToGround,3,0,"R0!",1.0,0) 
    EndIf 
 
    If TimeIntoInterval (0,60,Min) Then SR50ATMeasureFlag = TRUE 
 
    If SR50ATMeasureFlag = TRUE Then 
      'perform 11 Snow Depth Measurements Back to Back 
      For SR50Acount = 1 To 11 
        'Read the Snow depth from the SR50AT "M4!" command returns depth with 
        'quality and tempeature 
        SDI12Recorder (SR50AReturnValues(SR50Acount,1),3,0,"M4!",1.0,0) 
        SR50ADistSort(SR50Acount) = SR50AReturnValues(SR50Acount,1) 
      Next 
 
      SortSpa (SR50ASort(1),11,SR50ADistSort(1)) 
 
      'Find the Mean Snow depth Value 
      SR50A_SnowDepth_Meters =  SR50ASort(6) 
 
      'find the corresponding Quality and Temperature 
      For SR50Acount = 1 To 11 
        'Read the Snow depth from the SR50A "M4!" command returns depth with 
        'quality and tempeature 
        If SR50AReturnValues(SR50Acount,1) = SR50A_SnowDepth_Meters Then 
          SR50A_QualityVal = SR50AReturnValues(SR50Acount,2) 
          SR50A_AirTempC = SR50AReturnValues(SR50Acount,3) 
          ExitFor 
        EndIf 
      Next SR50Acount 
 
      SR50ATMeasureFlag = FALSE 
    EndIf 
 
    'Call Output Tables. 
    CallTable Diagnostics 
    CallTable SR50AT 
  
  NextScan 
EndProg 

 


